# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=python-minikerberos
_pkgname=minikerberos
pkgver=0.4.7
_pyver=3.13
pkgrel=1
epoch=1
pkgdesc='Kerberos manipulation library in pure Python.'
arch=('any')
url='https://pypi.org/project/minikerberos/#files'
license=('MIT')
depends=('python' 'python-asn1crypto' 'python-oscrypto' 'python-asysocks'
         'python-unicrypto')
makedepends=('python-setuptools')
source=("https://files.pythonhosted.org/packages/source/${_pkgname::1}/$_pkgname/$_pkgname-$pkgver.tar.gz")
sha512sums=('7c25da8cd84cc031b08bf1cf5ccb33308bfffe7d5c73f5f2120191f12612cfbc192148c8ea4011c192e91ff09c294a11ca5b4f5d6c5532a7fcc5ed88e1736e43')

build() {
  cd "$_pkgname-$pkgver"

  python setup.py build
}

package() {
  cd "$_pkgname-$pkgver"

  python setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build

  rm -rf "$pkgdir/usr/lib/python$_pyver/site-packages/tests"
}

